Les sommations

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Ce document est une mise à jour de ceux produits en 2000 et 2006. L'objectif principal de ce document est de transposer en Maple la notation Σ utilisée en mathématique.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
```

Le symbole de sommation

La lettre sigma majuscule Σ est une lettre de l'alphabet grec. L'utilisation de la lettre Σ a été introduite par le mathématicien Leonard Euler (1707-1783) afin de symboliser une addition de termes ayant un lien d'écriture entre-eux.
La somme de termes consécutifs $a_m + a_{m+1} + a_{m+2} + \ldots + a_{n-1} + a_n$ est symbolisée par l'expression $\sum_{k=m}^{n} a_k$ où $m \le n$.
La lettre k est appelée variable ou indice de sommation. Le référentiel de la variable k est l'ensemble des entiers $\mathbb{Z}$. L'égalité $k = m$ indique que la valeur initiale de la variable de sommation k est m, et la lettre n est la valeur finale que prend la variable k. Les autres valeurs que prend la variable k sont successivement les entiers consécutifs m+1, m+2, m+3 et ainsi de suite.

$$\sum_{k=m}^{n} a_k = a_m + a_{m+1} + a_{m+2} + \ldots + a_{n-1} + a_n$$

La macro-commande Sum de la bibliothèque principale permet de transposer une sommation indéfinie dans Maple.

```
> Sum(a[k],k=m..n);
```

$$\sum_{k=m}^{n} a_k \qquad \textbf{(2.1)}$$

Le choix de la lettre utilisée pour spécifier la variable de sommation est arbitraire. Les lettres, i, j, k, et l sont le plus souvent utilisées comme variables de sommation.

$$\sum_{k=m}^{n} a_k = \sum_{i=m}^{n} a_i = a_m + a_{m+1} + a_{m+2} + \ldots + a_{n-1} + a_n$$

Voici quelques exemples de sommation avec différentes variables de sommation exprimant la même somme.

```
> Sum(a[i],i=1..6)=sum(a[i],i=1..6);
  Sum(a[j],j=1..6)=sum(a[j],j=1..6);
  Sum(a[k],k=1..6)=sum(a[k],k=1..6);
```

$$\sum_{i=1}^{6} a_i = a_1 + a_2 + a_3 + a_4 + a_5 + a_6$$

$$\sum_{j=1}^{6} a_j = a_1 + a_2 + a_3 + a_4 + a_5 + a_6$$

$$\sum_{k=1}^{6} a_k = a_1 + a_2 + a_3 + a_4 + a_5 + a_6 \qquad \textbf{(2.2)}$$

La variable de sommation est en quelque sorte une variable muette ne servant qu'à développer la portée du symbole de sommation.

La valeur initiale de la variable de sommation peut être un entier quelconque, pas nécessairement l'entier 1.

$$\sum_{k=3}^{7} a_k = a_3 + a_4 + a_5 + a_6 + a_7$$

Exemple.

```
> Sum(a[k],k = 3 .. 7)=sum(a[k],k = 3 .. 7);
  Sum(b[k],k = -3 .. 4)=sum(b[k],k = -3 .. 4);
```

$$\sum_{k=3}^{7} a_k = a_3 + a_4 + a_5 + a_6 + a_7$$

$$\sum_{k=-3}^{4} b_k = b_{-3} + b_{-2} + b_{-1} + b_0 + b_1 + b_2 + b_3 + b_4 \qquad \textbf{(2.3)}$$

<u>REMARQUE</u>

Habituellement, en mathématique, la valeur initiale de la variable de sommation est un entier inférieur à la valeur finale de la variable, c'est-à-dire que l'on a ordinairement $m < n$:

$$\sum_{k=-7}^{-1} a_k = a_{-7} + a_{-6} + a_{-5} + a_{-4} + a_{-3} + a_{-2} + a_{-1}$$

Dans les cas inhabituels où la valeur initiale sera supérieure à la valeur finale, soit m > n, le mécanisme de la simplification automatique donne pour résultat la sommation

$$\sum_{k=m}^{n} a_k = -\left(\sum_{k=n+1}^{m-1} a_k \right)$$

Par exemple, soit la somme $\sum_{k=7}^{1} a_k$

```
> Sum(a[k],k = 7 .. 1)=sum(a[k],k = 7 .. 1);
```

$$\sum_{k=7}^{1} a_k = -a_2 - a_3 - a_4 - a_5 - a_6 \qquad \textbf{(2.4)}$$

Vous avez sans doute remarqué que la macro-commande `sum` possède une forme inactive qui permet de bien documenter la zone des résultats.

```
> Sum(a[k],k=1..5)=sum(a[k],k=1..5);
```

$$\sum_{k=1}^{5} a_k = a_1 + a_2 + a_3 + a_4 + a_5 \tag{2.5}$$

Notons, qu'avec la forme inactive de la macro-commande, l'afficheur présente la lettre Σ en gris. Il en sera toujours ainsi lorsque l'usager utilise la forme inactive d'une macro-commande `Maple`. Il en est de même avec les formes inertes `Diff`, `Int`, `Limit`, `Product`, etc. C'est particulièrement commode d'avoir cela dans la zone des résultats car cela permet, par simple examen visuel, de faire rapidement la différence entre la mise en forme imposée par la forme inactive et la présentation du résultat lorsque `Maple` n'a pu obtenir la simplification de la requête. En effet, lorsqu'il n'y a pas eu simplification par l'évaluateur, l'afficheur donne pour résultat la requête telle quelle comme un écho.

```
> Sum(a[k],k=m..n)=sum(a[k],k=m..n);
```

$$\sum_{k=m}^{n} a_k = \sum_{k=m}^{n} a_k \tag{2.6}$$

La réindexation d'une sommation

Nous avons vu que le choix de la lettre comme variable de sommation est arbitraire dans la notation Σ. Également, nous verrons maintenant que le choix même de la valeur initiale de cette variable est tout aussi arbitraire dans la symbolisation d'une sommation.

Soit la somme $\sum_{k=1}^{7} \frac{1}{2^k}$.

```
> Sum(1/2^k,k=1..7)=sum(1/2^k,k=1..7);
  Sum(1/2^(k-2),k=3..9)=sum(1/2^(k-2),k=3..9);
  Sum(1/2^(k+3),k=-2..4)=sum(1/2^(k+3),k=-2..4);
```

$$\sum_{k=1}^{7} \frac{1}{2^k} = \frac{127}{128}$$

$$\sum_{k=3}^{9} \frac{1}{2^{k-2}} = \frac{127}{128}$$

$$\sum_{k=-2}^{4} \frac{1}{2^{k+3}} = \frac{127}{128} \tag{3.1}$$

Nous pouvons effectivement réindexer une sommation sans changer la valeur de chaque terme et donc, de la somme. Alors, pour ne pas changer la valeur d'une
somme,

- si nous augmentons de r unités les valeurs initiale et finale de la variable de sommation, il faut diminuer de r unités chaque apparition de cette variable dans la portée
- si nous diminuons de r unités les valeurs initiale et finale de la variable de sommation, il faut augmenter de r unités chaque apparition de cette variable dans la portée.

$$\sum_{k=m}^{n} a_k = \sum_{k=m+r}^{n+r} a_{k-r} = \sum_{k=m-r}^{n-r} a_{k+r}$$

Gardons à l'esprit qu'il est préférable tout de même de symboliser une sommation le plus naturellement possible.

Règles de la sommation

L'évaluateur connaît bien sûr les règles élémentaires de la sommation.

RÈGLE 1

```
> Sum(a[k]+b[k],k=m..n)=expand(Sum(a[k]+b[k],k=m..n));
```

$$\sum_{k=m}^{n} \left(a_k + b_k\right) = \left(\sum_{k=m}^{n} a_k\right) + \left(\sum_{k=m}^{n} b_k\right) \qquad \textbf{(4.1)}$$

RÈGLE 2

```
> Sum(c*a[k],k=m..n)=expand(sum(c*a[k],k=m..n));
```

$$\sum_{k=m}^{n} c\, a_k = c \left(\sum_{k=m}^{n} a_k\right) \qquad \textbf{(4.2)}$$

RÈGLE 3

```
> Sum(a[k]+c,k=m..n)=value(expand(Sum(a[k]+c,k=m..n)));
```

$$\sum_{k=m}^{n} \left(a_k + c\right) = \left(\sum_{k=m}^{n} a_k\right) + c\,(n+1-m) \qquad \textbf{(4.3)}$$

RÈGLE 4

```
> Sum(c,k=m..n)=value(expand(Sum(c,k=m..n)));
```

$$\sum_{k=m}^{n} c = c\,(n+1-m) \qquad \textbf{(4.4)}$$

Remarque: Le facteur $(n+1-m)$ nous donne le nombre de termes dans une sommation.

Enfin, une sommation peut être scindée en deux sommations contiguës pour tout entier m tel que m < p < n :

RÈGLE 5

```
> Sum(b[k],k=m..n)=Sum(b[k],k=m..p)+Sum(b[k],k=p+1..n);
```

$$\sum_{k=m}^{n} b_k = \left(\sum_{k=m}^{p} b_k\right) + \left(\sum_{k=p+1}^{n} b_k\right) \qquad \textbf{(4.5)}$$

Réciproquement, deux sommations contiguës peuvent être unifiées en une seule sommation avec la macro-commande `combine` de la bibliothèque principale.

```
> Somme:=Sum(b[k],k=m..p)+Sum(b[k],k=p+1..n):
  Somme=combine(Somme);
```

$$\left(\sum_{k=m}^{p} b_k\right) + \left(\sum_{k=p+1}^{n} b_k\right) = \sum_{k=m}^{n} b_k \qquad \textbf{(4.6)}$$

Le mécanisme de la simplification automatique réduit un très grand nombre de sommations indéfinies par des formules connues. Par exemple, simplifions les sommes suivantes:

$$\sum_{k=1}^{n} k,\ \sum_{k=1}^{n} k^2 \quad \text{et} \quad \sum_{k=1}^{n} k^3.$$

```
> Sum(k,k = 1 .. n)=sum(k,k = 1 .. n);
  ``=normal(rhs(%));
  factor(%);
```

$$\sum_{k=1}^{n} k = \frac{(n+1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$
$$= \frac{1}{2} n^2 + \frac{1}{2} n$$
$$= \frac{n(n+1)}{2} \tag{4.7}$$

```
> Sum(k^2,k = 1 .. n)=sum(k^2,k = 1 .. n);
  ``=normal(rhs(%));
```

$$\sum_{k=1}^{n} k^2 = \frac{(n+1)^3}{3} - \frac{(n+1)^2}{2} + \frac{n}{6} + \frac{1}{6}$$
$$= \frac{1}{3} n^3 + \frac{1}{2} n^2 + \frac{1}{6} n \tag{4.8}$$

```
> Sum(k^3,k = 1 .. n)=sum(k^3,k = 1 .. n);
  ``=normal(rhs(%));
```

$$\sum_{k=1}^{n} k^3 = \frac{(n+1)^4}{4} - \frac{(n+1)^3}{2} + \frac{(n+1)^2}{4}$$
$$= \frac{1}{4} n^4 + \frac{1}{2} n^3 + \frac{1}{4} n^2 \tag{4.9}$$

Toutes les réponses précédentes peuvent, évidemment, être simplifiées soit

- sous la forme développée avec `normal(    ,expanded)` et
- sous la forme factorisée avec `factor`.

Par exemple:

```
> Sum(k^3,k = 1 .. n)=sum(k^3,k = 1 .. n);
  ``= normal(rhs(%));
  ``=factor(rhs(%));
```

$$\sum_{k=1}^{n} k^3 = \frac{(n+1)^4}{4} - \frac{(n+1)^3}{2} + \frac{(n+1)^2}{4}$$
$$= \frac{1}{4} n^4 + \frac{1}{2} n^3 + \frac{1}{4} n^2$$
$$= \frac{n^2 (n+1)^2}{4} \tag{4.10}$$

Pour obtenir la formule de la somme indéfinie

$\sum_{k=1}^{n} \left(3 k^4 - k^2 + 1\right)$, il n'est pas nécessaire de faire intervenir explicitement les propriétés du symbole de sommation. L'évaluateur les applique automatiquement.:

```
> Sum(3*k^4-k^2+1,k = 1 .. n)=sum(3*k^4-k^2+1,k = 1 .. n);
```

$$\sum_{k=1}^{n} \left(3 k^4 - k^2 + 1\right) = \frac{3 (n+1)^5}{5} - \frac{3 (n+1)^4}{2} + \frac{2 (n+1)^3}{3} + \frac{11 n}{15} - \frac{4}{15} + \frac{(n+1)^2}{2} \qquad \textbf{(4.11)}$$

```
> factor(%);
```

$$\sum_{k=1}^{n} \left(3 k^4 - k^2 + 1\right) = \frac{n \left(18 n^4 + 45 n^3 + 20 n^2 - 15 n + 22\right)}{30} \qquad \textbf{(4.12)}$$

Si on désire obtenir la décomposition d'une sommation à l'aide des propriétés du symbole de sommation, il faut utiliser la macro-commande `expand` sur la forme inactive de la la sommation.

```
> Sum(3*k^4-k^2+1,k = 1 .. n)=expand(Sum(3*k^4-k^2+1,k = 1 .. n));
```

$$\sum_{k=1}^{n} \left(3 k^4 - k^2 + 1\right) = 3 \left(\sum_{k=1}^{n} k^4\right) - \left(\sum_{k=1}^{n} k^2\right) + \left(\sum_{k=1}^{n} 1\right) \qquad \textbf{(4.13)}$$

La double sommation

En mathématique, on rencontre des sommations où interviennent deux symboles de sommation : $\sum_{i=1}^{n} \left(\sum_{j=1}^{m} x_i y_j\right)$.

La double sommation est une sommation d'une sommation. Dans les textes mathématiques, les doubles sommations s'écrivent habituellement sans parenthèses. Cela est la conséquence de la règle 1 (voir la sous-section suivante).

Dans la sommation imbriquée $\sum_{j=1}^{m} x_i y_j$ de la sommation précédente, les variables indicées x_i jouent un rôle de constante face à la sommation sur j. On a donc l'égalité :

$$\sum_{i=1}^{n} \left(\sum_{j=1}^{m} x_i y_j\right) = \sum_{i=1}^{n} x_i \left(\sum_{j=1}^{m} y_j\right).$$

```
> Sum(Sum(x[i]*y[j],j=1..m),i=1..n)=Sum(x[i]*Sum(y[j],j=1..m),i=1..n);
```

$$\sum_{i=1}^{n} \sum_{j=1}^{m} x_i y_j = \sum_{i=1}^{n} x_i \left(\sum_{j=1}^{m} y_j\right) \qquad \textbf{(5.1)}$$

Et nous écrivons

```
> ``=Sum(x[i],i=1..n)*Sum(x[j],j=1..m);
```

$$= \left(\sum_{i=1}^{n} x_i\right) \left(\sum_{j=1}^{m} x_j\right) \qquad \textbf{(5.2)}$$

Obtenons ce dernier développement à l'aide de `expand` appliquée directement sur la double sommation.

```
> Sum(Sum(x[i]*y[j],j=1..m),i=1..n)=expand(Sum(Sum(x[i]*y[j],j=1..m),i=
  1..n));
```

$$\sum_{i=1}^{n}\sum_{j=1}^{m} x_i y_j = \sum_{i=1}^{n} x_i \left(\sum_{j=1}^{m} y_j\right) \quad \textbf{(5.3)}$$

On rencontre donc rarement, en mathématique, la multiplication de deux sommations car il s'agit en fait d'une double sommation. La preuve est assez simple à faire. Illustrons plutôt ici la véracité de cette affirmation en développant selon les deux manières la double sommation $\sum_{i=1}^{4}\left(\sum_{j=1}^{2} x_i y_j\right)$.

Voici, d'une part, le développement en tant que sommation d'une sommation.

```
> Sum(Sum(x[i]*y[j],j=1..2),i=1..4)=sum(sum(x[i]*y[j],j=1..2),i=1..4);
```

$$\sum_{i=1}^{4}\sum_{j=1}^{2} x_i y_j = x_1 y_1 + x_1 y_2 + x_2 y_1 + x_2 y_2 + x_3 y_1 + x_3 y_2 + x_4 y_1 + x_4 y_2 \quad \textbf{(5.4)}$$

Et d'autre part, le développement en tant que multiplication de deux sommation.

```
> Sum(y[j],j = 1 .. 2)*Sum(x[i],i = 1 .. 4)=sum(y[j],j = 1 .. 2)*sum(x
  [i],i = 1 .. 4);
```

$$\left(\sum_{j=1}^{2} y_j\right)\left(\sum_{i=1}^{4} x_i\right) = (y_1 + y_2)(x_1 + x_2 + x_3 + x_4) \quad \textbf{(5.5)}$$

```
> expand(%);
```

$$\left(\sum_{j=1}^{2} y_j\right)\left(\sum_{i=1}^{4} x_i\right) = x_1 y_1 + x_1 y_2 + x_2 y_1 + x_2 y_2 + x_3 y_1 + x_3 y_2 + x_4 y_1 + x_4 y_2 \quad \textbf{(5.6)}$$

Ce qui montre que l'on a bien l'égalité $\sum_{i=1}^{4}\left(\sum_{j=1}^{2} x_i y_j\right) = \left(\sum_{j=1}^{2} y_j\right)\left(\sum_{i=1}^{4} x_i\right)$.

Règles de la double sommation

En ce qui concerne la double sommation, l'évaluateur reconnaît la règle suivante.

RÈGLE 1

Dans une double sommation, nous pouvons permuter les deux « Σ » : $\sum_{i=1}^{n}\left(\sum_{j=1}^{m} x_i y_j\right) = \sum_{j=1}^{m}\left(\sum_{i=1}^{n} x_i y_j\right)$.

```
> Sum(Sum(x[i]*y[j],j = 1 .. m),i = 1 .. n) = Sum(Sum(x[i]*y[j],i = 1 .
  . n),j = 1 .. m);
```

$$\sum_{i=1}^{n}\sum_{j=1}^{m} x_i y_j = \sum_{j=1}^{m}\sum_{i=1}^{n} x_i y_j \quad \textbf{(6.1)}$$

Illustrons la règle 1 par le développement limité suivant : $\sum_{i=1}^{2}\left(\sum_{j=1}^{4} x_i y_j\right)$.

```
> Sum(Sum(x[i]*y[j],j = 1 .. 4),i = 1 .. 2)=sum(sum(x[i]*y[j],j = 1 ..
  4),i = 1 .. 2);
```

$$\sum_{i=1}^{2}\sum_{j=1}^{4} x_i y_j = x_1 y_1 + x_1 y_2 + x_1 y_3 + x_1 y_4 + x_2 y_1 + x_2 y_2 + x_2 y_3 + x_2 y_4 \quad \textbf{(6.2)}$$

```
> Sum(Sum(x[i]*y[j],i = 1 .. 2),j = 1 .. 4)=sum(sum(x[i]*y[j],i = 1 ..
  2),j = 1 .. 4);
```

$$\sum_{j=1}^{4}\sum_{i=1}^{2} x_i y_j = x_1 y_1 + x_1 y_2 + x_1 y_3 + x_1 y_4 + x_2 y_1 + x_2 y_2 + x_2 y_3 + x_2 y_4 \quad \textbf{(6.3)}$$

Voici quatres autres règles de la double sommation.

RÈGLE 2

```
> Sum(Sum(a[i,j]+b[i,j],j = 1 .. m),i = 1 .. n) = expand(Sum(Sum(a[i,j]
  +b[i,j],j = 1 .. m),i=1..n));
```

$$\sum_{i=1}^{n}\sum_{j=1}^{m} \left(a_{i,j} + b_{i,j}\right) = \left(\sum_{i=1}^{n}\sum_{j=1}^{m} a_{i,j}\right) + \left(\sum_{i=1}^{n}\sum_{j=1}^{m} b_{i,j}\right) \quad \textbf{(6.4)}$$

RÈGLE 3

```
> Sum(Sum(k*a[i,j],j = 1 .. m),i = 1 .. n) = expand(Sum(Sum(k*a[i,j],j
  = 1 .. m),i = 1 .. n));
```

$$\sum_{i=1}^{n}\sum_{j=1}^{m} k a_{i,j} = \sum_{i=1}^{n} k \left(\sum_{j=1}^{m} a_{i,j}\right) \quad \textbf{(6.5)}$$

RÈGLE 4

```
> Sum(Sum(x[i],j = 1 .. m),i = 1 .. n) = expand(Sum(sum(x[i],j = 1 ..
  m),i = 1 .. n));
  Sum(Sum(y[j],j = 1 .. m),i = 1 .. n) = expand(sum(Sum(y[j],j = 1 ..
  m),i = 1 .. n));
```

$$\sum_{i=1}^{n}\sum_{j=1}^{m} x_i = m \left(\sum_{i=1}^{n} x_i\right)$$

$$\sum_{i=1}^{n}\sum_{j=1}^{m} y_j = n \left(\sum_{j=1}^{m} y_j\right) \quad \textbf{(6.6)}$$

RÈGLE 5

```
> Sum(Sum(k,j = 1 .. m),i = 1 .. n) = value(Sum(Sum(k,j = 1 .. m),i = 1
  .. n));
```

$$\sum_{i=1}^{n}\sum_{j=1}^{m} k = n\,m\,k \quad \textbf{(6.7)}$$

Macro-commandes `sum` et `add`

La macro-commande `sum` est la macro-commande à utiliser ***pour obtenir l'expression formelle d'une somme indéfinie***, qu'elle soit finie ou infinie.

Développons, sous la forme d'une égalité, la somme indéfinie $\sum_{k=1}^{n} 5^{k-1}$.

```
> Sum(5^(k-1),k = 1 .. n)=sum(5^(k-1),k = 1 .. n);
```

$$\sum_{k=1}^{n} 5^{k-1} = \frac{5^{n+1}}{20} - \frac{1}{4} \tag{7.1}$$

La somme indéfinie précédente est une somme finie. Le mécanisme de la simplification automatique de Maple reconnaît plusieurs sommations infnies.

```
> Sum((-1)^k*1/k,k=1..infinity)=sum((-1)^k*1/k,k=1..infinity);
```

$$\sum_{k=1}^{\infty} \frac{(-1)^k}{k} = -\ln(2) \tag{7.2}$$

Un tel résulat découle de l'étude des séries en générale, ce qui déborde le niveau du cours EED.

Dans certains cas, la formulation du résultat est exprimée avec des expressions mathématiques inconnues du niveau collégial.

```
> Sum(k/(k+1),k=0..n) = sum(k/(k+1), k=0..n);
```

$$\sum_{k=0}^{n} \frac{k}{k+1} = n + 1 - \Psi(n+2) - \gamma \tag{7.3}$$

Faisons connaissance avec le nombre γ et la fonction d'Euler Ψ (Psi) . Le nombre γ (gamma) est la constante d'Euler valant approximativement:

```
> gamma=evalf(gamma,20);
```

$$\gamma = 0.5772156649 \tag{7.4}$$

La fonction d'Euler Ψ (Psi) étant définie par $\frac{\mathrm{d}}{\mathrm{d}x} \ln(\Gamma(x))$ où $\Gamma(x) = \int_0^{\infty} \mathrm{e}^{-t} t^{x-1} \, \mathrm{d}t$.

Pas de panique !...Tous ces détails, le nombre γ, la fonction Ψ, c'est ici, seulement pour épater la galerie...Ce n'est pas des éléments mathématiques que l'on va aborder dans notre cours de compléments mathématiques. Sachez tout simplement, qu'en faisant des explorations avec le logiciel Maple, il est très probable de rencontrer des expressions mathématiques inconnues de notre niveau d'enseignement. Cela ne doit en aucun temps freiner votre goût de faire des explorations. Au contraire, ces "découvertes" devraient plutôt aiguiser votre curiosité.

Lorsqu'il s'agit d'obtenir la somme d'un nombre fini de valeurs numériques, c'est-à-dire lorsqu'on cherche explicitement une somme numérique plutôt qu'une formule, il est recommandé d'utiliser la macro-commande `add` même si la macro-commande `sum` pourrait donner explicitement la somme.

Calculons $\sum_{k=4}^{37} (3k^2 + 1)$.

```
> add(3*k^2+1,k = 4 .. 37);
```

$$52717 \tag{7.5}$$

Il est quand même à propos d'utiliser la forme inactive de `sum` afin de bien documenter la zone des résultats:

```
> Sum(3*k^2+1,k = 4 .. 37)=add(3*k^2+1,k = 4 .. 37);
```

(7.6)

$$\sum_{k=4}^{37} \left(3k^2+1\right) = 52717 \qquad (7.6)$$

Résumons. Pour opérer **symboliquement** une sommation indéfinie, on utilise **`sum`** tandis que pour opérer **numériquement** une sommation définie, on utilise **`add`**.
Sur les intervalles numériques `sum` et `add` donne la somme:

```
> add(k,k=1..12);
  sum(k,k=1..12);
```

$$78$$

$$78 \qquad (7.7)$$

Tandis que sur les intervalles symboliques, `add` est inopérante:

```
> add(k,k=1..n);
Error, unable to execute add
```

Tandis que `sum` donne un résultat formel.

```
> sum(k,k=1..n);
```

$$\frac{(n+1)^2}{2} - \frac{n}{2} - \frac{1}{2} \qquad (7.8)$$

En fait, la macro-commande `add` est une procédure définie à l'aide d'une structure de contrôle appelée *boucle*:

```
        add(Formule, k=m..n) <--> S := 0;
                                  Valeur := i;
                                  for i from m to n do S := S+Formule
                                  end do;
                                  i := Valeur;
                                  S; # Affichage du résultat
```

tandis que la macro-commande `sum` fait appel à un ensemble de résulats découlant de l'étude des polynômes. C'est ce qui explique son utilisation lorsqu'on effectue *un développement analytique d'une sommation.*

Exercices

No.1

À l'aide de la macro-commande `add`, obtener la somme des 100 premiers nombres impairs

$$1 + 3 + 5 + 7 + 9 + ... + 199$$

avec comme valeur initale de la variable de sommation,
a) la valeur 0
b) la valeur 1

No.2

À l'aide de la macro-commande `sum`, obtener la formule factorisée de la somme des n premiers nombres impairs consécutifs avec comme valeur initiale de la variable de sommation,
a) la valeur 0
b) la valeur 1

No.3

À l'aide de la macro-commande `sum`, vérifier que la somme des n premier termes donne,

(Vous devez touver le terme général de la sommation.)

a) $1 + 2 + 3 + 4 + 5 + \ldots \;=\; \dfrac{n\,(n+1)}{2}$

b) $1^2 + 2^2 + 3^2 + 4^2 + 5^2 + \ldots \;=\; \dfrac{n\,(n+1)\,(2\,n+1)}{6}$

c) $1^3 + 2^3 + 3^3 + 4^3 + 5^3 + \ldots \;=\; \dfrac{n^2\,(n+1)^2}{4}$

c) $\dfrac{1}{1\cdot2\cdot3} + \dfrac{1}{2\cdot3\cdot4} + \dfrac{1}{3\cdot4\cdot5} + \ldots \;=\; \dfrac{n\,(n+3)}{4\,(n+1)\,(n+2)}$

d) $1 + 7 + 13 + 19 + \ldots \;=\; n\,(3\,n-2)$

e) $5^0 + 5 + 5^2 + 5^3 + \ldots \;=\; \dfrac{5^n - 1}{4}$

No.4

Pour les exercices ci-dessous, documenter votre solution comme l'exemple suivant.

Exemple

Obtener la somme infinie $\dfrac{1}{3} + \dfrac{1}{15} + \dfrac{1}{35} + \dfrac{1}{63} + \dfrac{1}{99} + \dfrac{1}{143} + \ldots + \dfrac{1}{(2\,k-1)\,(2\,k+1)} + \ldots$ de deux manières différentes

– en posant directement $\sum_{k=1}^{\infty} a_k$ et

– en posant $\lim_{n\to\infty} \left(\sum_{k=1}^{n} a_k\right)$.

et vérifier si l'évaluateur a, dans chaque cas, donné un résultat égal.

Solution

```
> P:=k->1/((2*k-1)*(2*k+1));
```

$$P := k \mapsto \frac{1}{(2\cdot k-1)\cdot(2\cdot k+1)} \tag{8.4.1}$$

```
> seq(P(k),k=1..6);
```

$$\frac{1}{3}, \frac{1}{15}, \frac{1}{35}, \frac{1}{63}, \frac{1}{99}, \frac{1}{143} \tag{8.4.2}$$

```
> Sum(P(k),k=1..infinity)=sum(P(k),k=1..infinity);
```

$$\sum_{k=1}^{\infty} \frac{1}{(2k-1)(2k+1)} = \frac{1}{2} \tag{8.4.3}$$

```
> Sum(P(k),k=1..infinity)=Limit(Sum(P(k),k=1..n),n=infinity);
  ``=Limit(sum(P(k),k=1..n),n=infinity);
  ``=limit(sum(P(k),k=1..n),n=infinity);
```

$$\begin{aligned}\sum_{k=1}^{\infty} \frac{1}{(2k-1)(2k+1)} &= \lim_{n\to\infty} \sum_{k=1}^{n} \frac{1}{(2k-1)(2k+1)} \\ &= \lim_{n\to\infty} \left(-\frac{1}{4\left(n+\frac{1}{2}\right)} + \frac{1}{2}\right) \\ &= \frac{1}{2}\end{aligned} \tag{8.4.4}$$

Dans les deux cas, l'évaluateur a donné un même résultat.

a) $\frac{1}{3} + \frac{1}{8} + \frac{1}{15} + \frac{1}{24} + \frac{1}{35} + \ldots + \frac{1}{k(k+2)} + \ldots$

b) $\frac{1}{4} + \frac{1}{10} + \frac{1}{18} + \frac{1}{28} + \frac{1}{40} + \frac{1}{54} + \ldots + \frac{1}{k(k+3)} + \ldots$

c) $\frac{1}{2}+\frac{1}{3}+\frac{1}{8}+\frac{1}{30}+\frac{1}{144}+\frac{1}{840}+\ldots+\frac{k}{(k+1)!}+\ldots$

No.5

À l'aide de la macro-commande `add`, évaluer

a) $\sum_{i=1}^{4}\left(\sum_{j=1}^{4}\left(i^2-1\right)(j+1)\right)$

b) $\sum_{i=1}^{4}\left(\sum_{j=1}^{6}(ij+2i)\right)$

No.6

Montrer que $\left(\sum_{i=1}^{n}x_i\right)^2=\sum_{i=1}^{n}\left(\sum_{j=1}^{n}x_i x_j\right)$

No.7

Monter que $\left(\sum_{i=1}^{n}\left(x_i+1\right)\right)^2=\sum_{i=1}^{n}\left(\sum_{j=1}^{n}x_i x_j\right)+2n\left(\sum_{i=1}^{n}x_i\right)+n^2$

No.8

Similairement à ce qu'on fait avec le symbole Σ, la lettre π majuscule de l'alphabet grec, soit Π, est utlilisée pour symboliser une multiplication de facteurs consécutifs :

$\prod_{k=1}^{n}a_k = a_1 a_2 a_3 \ldots a_{n-1} a_n.$

```
> Product(a[k],k=1..n);
```

$$\prod_{k=1}^{n}a_k \tag{8.8.1}$$

```
> Product(a[k],k=1..7)=product(a[k],k=1..7);
```

$$\prod_{k=1}^{7}a_k=a_1 a_2 a_3 a_4 a_5 a_6 a_7 \tag{8.8.2}$$

Développer les multiplications suivantes. Documenter la zone des résultats sous la forme d'une égalité en employant la forme inerte de la macro-commande.

a) $\prod_{k=1}^{7}k$

b)

$$\prod_{k=3}^{9} \frac{(k+1)\,k}{2}$$

c) $\prod_{k=1}^{20} \left(\frac{k}{k+2}\right)^3$

d) $\prod_{k=1}^{20} \left(\frac{x}{x+2}\right)^3$

e) $\prod_{k=-5}^{5} (2k-15)$

f) $\prod_{k=-5}^{5} (2k\pi+1)$